
Visual Basic - Guía del Estudiante  Cap. 5
CONTROLES DE BUSQUEDA DE FICHEROS - PORTAPAPELES

MENÚS - MENUS EMERGENTES

CONTROLES DE BUSQUEDA DE FICHEROS Y DIRECTORIOS

· DriveListBox, DirListBox Y FileListBox.

Se obtienen directamente de la Caja de Herramientas, en los iconos siguientes:

· DriveListBox permite elegir una unidad de disco.

· DirListBox muestra los directorios existentes en un árbol.

· FileListBox muestra los ficheros existentes en un determinado directorio.

Tal como indica su nombre, estos controles son listas. Repase el control ListBox para recordar sus propiedades.

DriveListBox

Este control permite elegir una unidad de disco válida en tiempo de ejecución. Utilice este control para mostrar una lista de todas las unidades válidas del sistema de un usuario. Puede crear cuadros de diálogo que permitan al usuario abrir un archivo de una lista en cualquier unidad disponible.



  

DriveListBox  desplegado y sin desplegar   Name (nombre) por defecto  List1

Posee las propiedades típicas de cualquier control VB respecto a su geometría y color. Las Propiedades mas destacables de este control son : Drive, List, ListCount y ListIndex



Drive devuelve el nombre de la unidad de disco seleccionada :


Variable = List1.Drive 
Variable = a :    (ó c :[MS-DOS_6]   ó  D ) 



Este control se trata de una lista como su nombre indica, y posee las propiedades List, ListCount y ListIndex para permitir al usuario tener acceso a elementos de la lista. 

List (n) devuelve el texto que define a la unidad colocada en el lugar n. El número n comienza por 0.  Por ejemplo,  en la figura anterior,  List1.list(1) devolverá c :[MS-DOS_6]

ListCount proporciona información acerca del número de unidades de disco disponibles (En el ejemplo, List1.ListCount será 3,  que corresponde a los tres discos disponibles - disco C :, disco A :  y disco D : )

ListIndex devuelve el índice del disco seleccionado en la lista (0 para el primero, n-1 para el último)

Utilizando el control DriveListBox en combinación con  DirListBox y FileListBox, podrá realizar un explorador de ficheros en cualquier disco de su ordenador.

DirListBox. 

Presenta un cuadro de la siguiente forma:

donde puede verse el árbol completo de un directorio. El DirListBox solamente presenta directorios. No presenta ficheros.

Aparte de las propiedades geométricas de todos los controles, el DirListBox presentas las siguientes propiedades particulares :

Path
Devuelve o establece la ruta de acceso actual. No está disponible en tiempo de diseño. 

Sintaxis
Nombre.Path [= pathname]

donde

Nombre 
es el nombre del DirListBox
pathname
Una expresión de cadena que evalúa en el nombre de vía de acceso del directorio por defecto del DirListBox..

El valor de la propiedad Path es una cadena que indica una ruta de acceso, como C:\WINDOWS\SYSTEM.  El valor predeterminado es la ruta de acceso actual 

Cuando se cambia la propiedad Path en tiempo de ejecución se genera un evento Change.

Nota   El valor que devuelve Path es distinto del de List (ListIndex). Path devuelve siempre un mismo camino, bien el que se le haya impuesto mediante esta misma propiedad Path, y si no se le ha impuesto con anterioridad, el que tenga por defecto. List (ListIndex) devuelve el camino completo del directorio seleccionado, y, como era de esperar, cambia cada vez que cambiamos la selección del directorio.

List, ListCount y ListIndex
Al igual que cualquier lista, el DirListBox tiene las propiedades List, ListCount y ListIndex.

ListIndex devuelve el valor del índice del directorio seleccionado. El valor devuelto es cero cuando está seleccionado el primer directorio de los situados al final del árbol, 1 para el siguiente,  etc. Toma el valor -1 cuando se selecciona el directorio inmediatamente superior al último nivel, el valor -2 cuando se selecciona el directorio situado 2 niveles jerárquicos por encima del último nivel, etc.

ListCount devuelve un valor numérico igual al número de directorios existentes en el último nivel, independientemente de los directorios que tuviesen por arriba.

List (n)  devuelve una cadena de caracteres con el camino completo del elemento cuyo orden sea el señalado entre paréntesis.  Si en vez de n colocamos el ListIndex nos devolverá, logicamente, el camino del elemento seleccionado.

FileListBox

Esta lista nos presenta los ficheros existentes en un directorio.





Como se puede apreciar, cuando el número de ficheros supera el número que puede presentar, aparecen automáticamente las flechas de scroll verticales.

Esta lista de ficheros tiene las siguientes propiedades destacables :

Path
Devuelve o establece la ruta de acceso de su directorio actual. No está disponible en tiempo de diseño. Este directorio puede ser el que se le haya impuesto mediante esta propiedad Path, o el que tenga por defecto. (El directorio por defecto sería el del ejecutable de la aplicación)

Si le ponemos el nombre por defecto, List1, para todos los ejemplos :

variable = List1.Path          Variable = C:\Archivos de programa\Microsoft Visual Basic32
    (p. e.)

ListCount

Devuelve el número total de ficheros que hay en ese directorio.

Variable = List1.ListCount
Variable = 19         (p. e.)

 ListIndex
Devuelve un número con el índice del elemento seleccionado. Como todas las listas, devuelve -1 si no hay ningún elemento seleccionado. El primer elemento de una lista tiene el índice = 0.

Filename
Devuelve el nombre del fichero seleccionado.

List (n) 

Devuelve el nombre del fichero que ocupa el lugar n en la lista.

Variable = List1.List (5)

Variable = nombre del fichero que ocupa el 5º lugar en la lista.

Si en lugar de n ponemos el valor de ListIndex :

variable = List1.List (List1.ListIndex)
obtendremos el nombre del fichero seleccionado.

Observe que en este último caso, el resultado es el mismo que el de  List1.Filename
 Veamos otras propiedades del FileListBox

Pattern

Devuelve o establece un valor que indica los nombres de archivo mostrados en un control FileListBox en tiempo de ejecución.


Sintaxis
objeto.Pattern [= valor]

La sintaxis de la propiedad Pattern consta de las siguientes partes:

objeto : Nombre del Control FileListBox 

valor: Expresión de cadena que indica una especificación de archivo, como *.* o *.FRM. La cadena predeterminada es *.*, que obtiene una lista de todos los archivos. Además de utilizar caracteres comodín, también puede especificar varios modelos, separándolos 
con caracteres punto y coma (;). Por ejemplo, con *.EXE; *.BAT se obtendría una lista con todos los archivos ejecutables y todos los 
archivos por lotes de MS-DOS.

Comentarios

La propiedad Pattern juega un papel clave en el diseño de las funciones de examen y tratamiento de archivos de una aplicación. Puede utilizar Pattern en combinación con otras propiedades de control de archivos para proporcionar al usuario formas de explorar archivos o grupos de archivos similares. Por ejemplo, en una aplicación dedicada a iniciar otros programas, podría hacer que sólo apareciesen los archivos .EXE en el cuadro de lista de archivos (*.EXE). 

Al modificar el valor de la propiedad Pattern se genera un evento PatternChange.

Archive, Hidden, Normal, System

Devuelven o establecen un valor que determina si un control FileListBox presenta los archivos con los atributos Archive, Hidden, Normal o System.

El atributo Archive de un fichero indica si se ha variado ese fichero después de la última copia de seguridad.

El atributo Hidden indica si el fichero se trata de un fichero oculto.

El atributo Normal indica que el fichero es un fichero “normal” (No está oculto ni es de sistema).

El atributo System de un fichero indica que el fichero es un fichero de sistema.

Sintaxis


objeto.Archive [= booleano]


objeto.Hidden  [= booleano]


objeto.Normal  [= booleano]


objeto.System  [= booleano]

Las sintaxis de la propiedades Archive, Hidden, Normal y System tiene la siguiente partes:


objeto :Nombre del FileListBox
booleano :Una expresión booleana (True / False) que especifica el tipo de archivos presentados.

El valor booleano puede presentar los siguientes valores :

True :(Predeterminado para Archive y Normal) Presenta archivos con el atributo de la propiedad en el control FileListBox.

False : (Predeterminado para Hidden y System) Presenta archivos sin el atributo de la propiedad en el control FileListBox.

Use estas propiedades para especificar el tipo de archivos a presentar en un control FileListBox, basándose en los atributos de archivos estándar usados en el entorno operativo. La definición de estas propiedades en el código en tiempo de ejecución hace que el control FileListBox presente sólo aquellos archivos con los atributos especificados.

Por ejemplo, en una operación de buscar y reemplazar se podrían presentar solamente archivos del sistema estableciendo la propiedad System a True y las demás propiedades a False. También, como parte del procedimiento de copia de seguridad, se podría establecer la propiedad Archive a True para relacionar solamente aquellos archivos modificados a partir de la última copia de seguridad.

PROCEDIMIENTOS del FileListBox


Click
Change
DragDrop   
DragOver
GotFocus


KeyDown
KeyPress
KeyUp


SENTENCIAS DE MANEJO DE DIRECTORIOS Y FICHEROS

El Directorio Actual

Se llama directorio Actual al directorio que está actualmente en uso, bien porque haya sido forzado el control correspondiente a tener ese directorio en su propiedad Path, bien porque, al no haberse forzado ninguno, utilice el directorio por defecto.

El directorio por defecto es el último que se ha seleccionado por algún procedimiento. Por ejemplo, al arrancar una aplicación, el directorio por defecto será el mismo donde se encuentra el ejecutable. Pero si a lo largo de la aplicación cambiamos el directorio, seleccionando otro mediante un CommonDialog, ese directorio es el que queda por defecto. Asegúrese en sus aplicaciones que siempre elige la propiedad Path antes de presentar ficheros o directorios a través de los controles anteriores, para evitar que pueda mostrarle ficheros de un directorio no deseado.

Puede comprobar cual es el directorio actual. La función CurDir nos devuelve el directorio actual.

Ejemplo :  Introduzca una línea de código con la siguiente sentencia

Variable = CurDir
Al ejecutarla la variable Variable contendrá la información :

Variable = C:\Archivos de programa\Microsoft Visual Basic32
    (p. e.)

Puede darse el caso de que desee conocer el directorio actual de varias unidades de disco. Por ejemplo, puede estar trabajando en el directorio C:\CursoVB\Ejemplos y ser este el directorio actual en el disco C :, y tener un disquete en la unidad A : en el que la última vez que trabajó fue en el directorio A :\ApVB\Tema1 y ese es su directorio por defecto en el disco A. Para conocer el directorio actual del disco C : no necesitó especificar la unidad, ya que era precisamente la unidad C la unidad actual. Para conocer el directorio actual de la unidad A emplearía la expresión :

Variable = CurDir (“A”)    Variable tomará entonces el valor   Variable = “A :\ApVB\Tema1”

Decíamos que el disco C era la unidad actual. También podemos cambiar la unidad actual, mediante la sentencia ChDrive seguida de una letra (entre comillas dobles)

Ejemplo 
ChDrive “A”

cambia la unidad actual a la unidad A.    Observe que el 




parámetro A no tiene que llevar los dos puntos  ( : ), aunque si lo 




desea, puede ponerlos. También es indiferente que la letra sea 




mayúscula o minúscula.

También podemos forzar a que un directorio sea el directorio por defecto. La sentencia que tendremos que usar en ChDir.

Ejemplo :  ChDir “C :\Windows\System”  cambiaría el directorio actual al especificado.

Podemos incluso crear un directorio. Usaremos para ello la sentencia MkDir. Esta sentencia crea el directorio con el nombre indicado, sobre el directorio actual.

Si el directorio actual es el C:\VB\Icons  y ejecutamos la sentencia :

MkDir “Varios”      crearíamos el directorio   C:\VB\Icons\Varios

Podemos incluso eliminar un directorio. La sentencia para eliminar un directorio es RmDir

¡ Cuidado !
 Visual-Basic no le advertirá que puede borrar algo que no desea. Asegúrese muy bien mediante un aviso al usuario cada vez que elimine un directorio.
Vamos a ver ahora como podemos ver los nombres de los ficheros existentes en un determinado directorio. Muy sencillo, con un FileListBox. Efectivamente, así podría ser para presentarlos en pantalla.  Sin embargo, si deseamos imprimir la lista de ficheros existentes, guardarla en un fichero, etc., tenemos una forma que facilita esta labor. Igual que en DOS. La función Dir

A esta función le debemos añadir como parámetro el criterio de búsqueda, criterio formado por los caracteres o comodines que deseemos. Dir devuelve solamente el nombre de un fichero, por lo que tendremos que repetirla tantas veces como ficheros tenga el directorio. Pero solamente tendremos que introducir el criterio de búsqueda la primera vez.

En el siguiente ejemplo, sacaremos por impresora los nombres de los ficheros existentes en el directorio actual : (Este código lo meteremos en el procedimiento click de un botón de comando)

fichero = Dir("*.*")

‘Le decimos que imprima todos (*.*)

Do Until fichero = ""
‘Hasta que fichero sea una cadena de longitud cero..

Printer.Print  fichero
‘Imprime la variable fichero en la impresora

fichero = Dir


‘Toma el nombre de otro fichero

Loop 


‘Vuelve a Do Until .....


Printer.EndDoc

‘Termina la impresión y saca la hoja de la impresora

Ya sabemos sacar el directorio. Vamos a ver que otras cosas se pueden hacer con ficheros.

Función Kill.  A poco Inglés que sepa se habrá dado cuenta que esta función sirve para quitarse algo de enmedio. Esta función borra un fichero del disco.


Kill “C:\CursoVB\Ejemplos\Ejemplo1.txt” 

borra el fichero Ejemplo1.txt, que se encuentra en el disco C, directorio \ CursoVB\Ejemplos

Función FileCopy. Copia un fichero. La sintaxis de esta función es la siguiente :


FileCopy Origen, Destino

Si tenemos un TextBox llamado Text1 donde vamos a introducir el nombre (con su Path) del fichero origen, otro TextBox donde introduciremos el nombre (con su Path) que queremos dar a la copia, para copiar ese fichero ejecutaremos la sentencia :


FileCopy Text1.Text, Text2.Text

Si queremos copiar un fichero del disco A, en el directorio CursoVB  y de nombre Ejemplo1 al disco y directorio C:\CursoVB\Ejemplos y con el mismo nombre que tenía en el disco A, es decir, Ejemplo1, ejecutaremos la sentencia :


FileCopy “A :\CursoVB\Ejemplo1” , “C:\CursoVB\Ejemplos\Ejemplo1”

Observe la necesidad de dobles comillas para especificar que es, precisamente lo escrito en el código, lo que indica el origen y el destino. En el caso anterior no llevaba dobles comillas, debido a que era el contenido de un TextBox lo que se usaba para indicar el origen y el destino.

La instrucción Name cambia el nombre de un fichero. No solamente puede cambiar un nombre, puede incluso moverlo de sitio, pero siempre en la misma unidad de disco.

La sintaxis de Name es la siguiente :


Name nombrerutaantiguo As nombrerutanuevo

Donde nombrerutaantiguo es una expresión de cadena que indica el nombre del fichero (con su path) al que se le quiere cambiar el nombre, y nombrerutanuevo es una expresión de cadena que especifica el nombre nuevo y, si se desea mover de directorio, la nueva ubicación del archivo y la unidad de disco. El nombre y ubicación especificados en nombrerutanuevo no pueden ser los de un archivo existente. 

Tanto nombrerutanuevo como nombrerutaantiguo deben encontrarse en la misma unidad de disco. Si la ruta de acceso especificada en nombrerutanuevo existe y es distinta de la ruta de acceso especificada en nombrerutaantiguo, la instrucción Name coloca el archivo en el nuevo directorio o carpeta y cambia el nombre del archivo, de ser necesario. Si nombrerutanuevo y nombrerutaantiguo tienen distintas rutas de acceso y el mismo nombre de archivo, Name coloca el archivo en la nueva ubicación, sin cambiar el nombre del archivo. Por medio de Name se puede mover un archivo de un directorio o carpeta a otro, pero no se puede mover un directorio o carpeta.

Si se utiliza Name con un archivo abierto se produce un error. Antes de cambiar el nombre de un archivo, se debe cerrar.

En el ejemplo, vamos a cambiar el nombre del fichero C:\CursoVB\Ejemplos\Ejemplo1 por el de C :\CursoVB\Ejemplos\Prueba1


Name “C:\CursoVB\Ejemplos\Ejemplo1”  As “C :\CursoVB\Ejemplos\Prueba1”

En el siguiente, aparte de cambiarlo de nombre, le cambiamos de directorio.


Name “C:\CursoVB\Ejemplos\Ejemplo1”  As “C :\CursoVB\Ejercic\Prueba1”

Existe una función que le permite conocer el volumen de un fichero : FileLen. Mediante esta instrucción podrá saber, por ejemplo, si un fichero le va a caber en lo que le queda libre de un disquete, el tiempo previsto para enviarlo por módem, etc.

La sintaxis es la siguiente :


Variable = FileLen("C:\CursoVB\Ejemplos\Ejemplo1")

Variable tomará un valor numérico igual al número de bytes del fichero.

Otra función para el tratamiento de ficheros es FileDateTime, que le permite conocer la fecha y hora de creación o de la última modificación de un fichero.

La sintaxis es, para el caso del ejemplo :


Variable = FileDateTime ("C:\CursoVB\Ejemplos\Ejemplo1")

Variable puede tomar un valor como el siguiente :  Variable = 13/02/97 14:59:20

(Observe que de esta función se obtiene una variable de cadena.) Puede aprovechar esta función para borrar o hacer Backup de determinados ficheros que tengan mas de un tiempo de permanencia, sustituir ficheros por versiones mas recientes, etc.

Podemos obtener y modificar los atributos de un fichero. Los atributos de un fichero se refieren a si son de solo lectura, ficheros ocultos, de sistema, si ha sido modificado desde la última vez que se ha realizado un Backup, 

Para conocer los atributos de un fichero, utilizaremos la función GetAttr. Esta función devuelve un número, que representa la suma de los atributos de un archivo, directorio o carpeta o una etiqueta de volumen.

Sintaxis
Variable = GetAttr (nombreruta)

Donde nombreruta es una expresión de cadena que especifica un nombre de archivo, con su Path y unidad de disco. 

El valor devuelto por GetAttr es un número, igual a la suma de los siguientes valores de atributos:

Valor
Constante

Descripción

0
vbNormal

Normal.

1
vbReadOnly

Sólo lectura.

2
vbHidden

Oculto.

4
vbSystem

Archivo de sistema.

16
vbDirectory

Directorio o carpeta.

32
vbArchive

El archivo ha sido modificado después de efectuar la última copia 




de seguridad.

Si la expresión nombreruta representa un directorio, se lo indicará devolviendo el valor 16.

Para cambiar un atributo utilizaremos la función SetAttr. La sintaxis de esta función es la siguiente

Sintaxis
SetAttr nombreruta , atributos

Donde nombreruta es una expresión de cadena que especifica un nombre de archivo, con su Path y unidad de disco, y atributos es una o varias de las constantes o valores anteriores, cuya suma especifica los atributos de archivo. 

Dado que el valor 16 indica que lo expresado en nombreruta es un directorio, no puede pretenderse darle a ese directorio atributos que solamente se pueden aplicar a un fichero, ni puede pretender darle a un fichero un valor en el que entre el 16 como sumando. Estos casos le producirán un error en tiempo de ejecución. 

Si pretende establecer los atributos de un archivo abierto, se producirá un error en tiempo de ejecución. 

Existe una función, que no tiene nada que ver con la anterior, aunque puede parecerlo por su parecido semántico y que también devuelve atributos, pero de otro tipo. Esta función es FileAttr

Función FileAttr

Devuelve el modo de apertura de un archivo. Logicamente debe referirse a un archivo abierto en el momento de ejecutar esta función. Vea para mas detalles la sentencia Open.

Sintaxis
variable = FileAttr (numerocanal , tipoinformación)

Donde numerocanal es el número del canal de apertura de ese archivo, y tipoinformación es el tipo de información que deseamos de ese fichero. Si tipoinformación = 1 nos devuelve el modo de apertura del fichero, según la siguiente lista


Valor
Modo del archivo


1
Entrada


2
Salida


4
Aleatorio


8
Añadir


32
Binario

La Ayuda de VB dice que tipoinformación puede ser =2 para que FileAttr devuelva información sobre el selector de archivos del sistema operativo. Algunas veces la información de ayuda de VB promete cosas que, sin dudar de que sean ciertas, son difíciles de comprobar.

Esta función no tiene una gran aplicación práctica, ya que comprueba la forma en la que hemos abierto un archivo. Información que conocemos desde el mismo momento de abrirlo.

Visual Basic - Guía del Estudiante  Cap. 5 Cont.
EL PORTAPAPELES.  EL OBJETO CLIPBOARD

OBJETO CLIPBOARD

Proporciona acceso al Portapapeles del sistema.

Sintaxis

Clipboard
El objeto Clipboard se usa para manipular el texto y los gráficos del Portapapeles. Este objeto puede usarse para permitir que el usuario corte, copie y pegue texto y gráficos en la aplicación.

El objeto Clipboard puede contener varias unidades de datos siempre y cuando cada una tenga un formato diferente. Por ejemplo, se puede usar el método SetData para poner un mapa de bits en el Clipboard con el formato CF_BITMAP y después usar el método SetText con el formato CF_TEXT para poner texto en el Clipboard. Después se puede usar el método GetText para recuperar el texto o el método GetData para recuperar el gráfico. Los datos del Clipboard se pierden cuando se colocan otros datos con el mismo formato en el Clipboard desde el código o mediante un comando de menú.

SetText
Pone una cadena de texto en el objeto Clipboard usando el formato del objeto Clipboard especificado. No acepta argumentos con nombre.

Sintaxis

Clipboard.SetText datos, formato

datos
Requerido. Cadena de datos que se va a colocar en el Portapapeles.

formato
Opcional. Una constante o valor que especifica uno de los formatos del Portapapeles reconocidos por Visual Basic, como se describe a continuación :

Los valores de formato son:

Constante
Valor

Descripción

vbCFLink
&HBF00
Información de conversación DDE

vbCFRTF
&HBF01
Formato de texto enriquecido

vbCFText
1

(Predeterminado) Texto

Estas constantes se enumeran en la biblioteca de objetos de Visual Basic (VB) del Examinador de objetos.

SetData

Pone una imagen en el objeto Clipboard usando el formato gráfico especificado. No acepta argumentos con nombre.

Sintaxis

Clipboard.SetData datos, formato

datos

Requerido. El gráfico que se va colocar en el objeto Clipboard. 

Importante.  Ver NOTA1 mas adelante

formato 
Opcional. Una constante o valor que especifica uno de los formatos del 

objeto Clipboard reconocidos por  Visual Basic, como se describe mas adelante. Si formato se omite, SetData determina el formato gráfico automáticamente.

Los valores de formato son:

Constante
Valor
Descripción

vbCFBitmap
2
Mapa de bits (archivos .BMP)

vbCFMetafile
3
Metarchivo (archivos .WMF)

vbCFDIB
8
Mapa de bits independiente del dispositivo (DIB)

vbCFPalette
9
Paleta de colores

Estas constantes se enumeran en la biblioteca de objetos de Visual Basic (VB) del Examinador de objetos.

El gráfico que se va colocar en el objeto Clipboard se define mediante la función LoadPicture o la propiedad Picture de los objetos Form, Image o PictureBox.

NOTA 1. Esta información está obtenida de la ayuda de VB. Observe que dice que el gráfico puede ser la propiedad Picture de un Form, Image o PictureBox.  

NO pretenda hacer esto

Clipboard.SetData Picture1.Picture

sino ESTO


Clipboard.SetData Picture1.Image

NOTA 2.  También existe un error en la forma de meter una imagen en el Portapapeles con el método Load Picture. La información dice que se introducirá de la siguiente forma :


Clipboard.SetData LoadPicture(“C:FIESTA.BMP)

Cuando vaya a introducir un fichero de mapa de bits (.BMP) ES NECESARIO poner que el fichero es del tipo 8 :


Clipboard.SetData LoadPicture(“C:FIESTA.BMP), 8

ya que si no, el Portapapeles contendrá solamente un cuadro negro. Si introduce otro tipo de ficheros, ( .WMF) no es necesario poner el número de tipo de fichero.

(Observe que esto NO coincide con lo que aporta la información de VB.)

GetData

Devuelve un gráfico del objeto Clipboard. No acepta argumentos con nombre.

Sintaxis

Clipboard.GetData (formato)

formato 

Opcional. Una constante o valor que especifica el formato gráfico de Clipboard, como se describe mas adelante. La constante o valor debe ir entre paréntesis. Si formato es 0 o se omite, GetData usa automáticamente el formato apropiado.

Los valores de formato son:

Constante
Valor
Descripción

vbCFBitmap
2
Mapa de bits (archivos .BMP)

vbCFMetafile
3
Metarchivo (archivos .WMF)

vbCFDIB
8
Mapa de bits independiente del dispositivo (DIB)

vbCFPalette
9
Paleta de colores

Si en el objeto Clipboard no hay ningún gráfico que coincida con el formato esperado, no se devuelve nada. Si en el objeto Clipboard sólo hay una paleta de colores, se crea un DIB de tamaño mínimo (1 x 1).

GetText

Devuelve una cadena de texto del objeto Clipboard. No acepta argumentos con nombre.

Sintaxis

Clipboard.GetText (formato)

formato

Opcional. Un valor o constante que especifica el formato del objeto Clipboard, como se describe mas adelante. La constante o valor debe ir entre paréntesis.

Los valores de formato son:

Constante
Valor
Descripción

vbCFLink
&HBF00
Información de conversación DDE

vbCFText
1

(Predeterminado) Texto

Si en el objeto Clipboard no hay ninguna cadena de texto que coincida con el formato esperado, se devuelve una cadena vacía ("").

Paste

Copia datos desde el Portapapeles del sistema en un control contenedor OLE.

Sintaxis

objeto.Paste

Objeto es el nombre del Panel Contenedor OLE donde se quiere depositar el contenido del Portapapeles.

Comentario

Para usar este método, defina la propiedad OLETypeAllowed y después verifique el valor de la propiedad PasteOK. No se puede pegar con éxito a menos que PasteOK devuelva True.

Si se ha ejecutado el método Paste, la propiedad OLEType es vbOLELinked (0) o vbOLEEmbedded (1). Si no se ha ejecutado el método Paste, la propiedad OLEType es vbOLENone (3).

Este método puede usarse para implementar un comando Edición Pegar en un menú.

Si el valor de la propiedad PasteOK es True y Visual Basic no puede pegar el objeto, el control contenedor OLE elimina cualquier objeto existente en el control.

PasteOK

Devuelve un valor que determina si el contenido del Portapapeles del sistema se puede pegar en el control contenedor OLE.

Sintaxis

objeto.PasteOK

Objeto es el nombre del contenedor OLE donde se quiere pegar el contenido del portapapeles.

Comentario

Cuando el valor de esta propiedad es True, el contenido del Portapapeles del sistema se puede pegar en el control contenedor OLE.

La propiedad OLETypeAllowed se usa para especificar el tipo del objeto (vinculado o incrustado) que se quiere pegar en el control contenedor OLE. Una vez que un objeto se ha pegado en el control contenedor OLE, se puede verificar el valor de la propiedad OLEType para determinar el tipo de objeto que ha sido creado.

Esta propiedad puede usarse si se quiere que la aplicación implemente un comando Pegar en un menú Edición. Si PasteOK es False, el comando del menú se deshabilita; si no, puede habilitarse. Los comandos de menú se habilitan y se deshabilitan estableciendo su propiedad Enabled a True o False, respectivamente.

Los objetos se pegan en el control contenedor OLE con el método Paste.

Para ofrecer mayor flexibilidad al usuario, presente un cuadro de diálogo Pegado especial cuando el usuario elija el comando Edición Pegar

. (Establezca OLETypeAllowed = 2 y después use el método PasteSpecialDlg). Cuando se presenta este cuadro de diálogo, el objeto del Portapapeles del sistema se pega basándose en las selecciones del usuario en el cuadro de diálogo.

GetFormat

Devuelve un entero que indica si un elemento del objeto Clipboard coincide con un formato especificado. No acepta argumentos con nombre.

Sintaxis

Clipboard.GetFormat (formato)

formato

Requerido. Un valor o constante que especifica el formato del objeto Clipboard, como se describe mas adelante. La constante o valor debe ir entre paréntesis.

Los valores de formato son:

Constante
Valor
Descripción

vbCFLink
HBF00
Información de conversación DDE

vbCFText
1
Texto

vbCFBitmap
2
Mapa de bits (archivos .BMP)

vbCFMetafile
3
Metarchivo (archivos .WMF)

vbCFDIB
8
Mapa de bits independiente del dispositivo (DIB)

vbCFPalette
9
Paleta de colores

El método GetFormat devuelve True si algún elemento del objeto Clipboard coincide con el formato especificado. Si no, devuelve False.

Para los formatos vbCFDIB y vbCFBitmap, la paleta de colores que se encuentre en el Clipboard se usa para presentar el gráfico.

Clear

Borra el contenido del Portapapeles del sistema.

Sintaxis

Clipboard.Clear


Con Clear se borra todo el contenido del portapapeles, texto y gráficos.

Copy

Copia el objeto de un control contenedor OLE al Portapapeles del sistema.

Sintaxis

objeto.Copy

Objeto es el nombre del contenedor OLE cuyo contenido queremos copiar

Comentario

Cuando se copia un objeto en el Portapapeles del sistema, todos los datos y la información de vinculación asociados con el objeto se copian en el Portapapeles del sistema. En el Portapapeles del sistema se pueden copiar objetos vinculados y objetos incrustados.

Este método puede usarse para implementar un comando Edición Copiar en un menú.

EJERCICIO

Se propone el siguiente ejercicio, donde pueden verse todas las posibilidades del Portapapeles.

Cree un formulario de la siguiente forma :





El gráfico que tiene el Picture de la izquierda puede ser cualquiera.

El código de este formulario es el siguiente :

Option Explicit

Private Sub Check1_Click()

End Sub

Private Sub Command1_Click()

Clipboard.SetData Picture1.Image

End Sub

Private Sub Command2_Click()

Dim ClpFmt As Integer

Dim Msg As String

On Error Resume Next    ' Configura el controlador de errores.

    If Clipboard.GetFormat(vbCFText) Then ClpFmt = ClpFmt + 1

    If Clipboard.GetFormat(vbCFBitmap) Then ClpFmt = ClpFmt + 2

    If Clipboard.GetFormat(vbCFDIB) Then ClpFmt = ClpFmt + 4

    If Clipboard.GetFormat(vbCFMetafile) Then ClpFmt = ClpFmt + 8

    Select Case ClpFmt

        Case 1

            Msg = "El Portapapeles sólo contiene texto."

        Case 2

            Msg = " El Portapapeles sólo contiene un archivo de mapa de bits."

        Case 3

           Msg = " El Portapapeles contiene texto y un archivo de mapa de bits."

        Case 5

            Msg = " El Portapapeles contiene texto y un archivo DIB."

        Case 8

            Msg = " El Portapapeles contiene solamente un archivo MetaFile."

        Case 9

            Msg = " El Portapapeles contiene texto y un archivo MetaFile."

Case Else

            Msg = "No hay nada en el Portapapeles."

    End Select

Label1.Caption = Msg ' Muestra el mensaje.

End Sub

Private Sub Command3_Click()

Clipboard.Clear

End Sub

Private Sub Command4_Click()

Clipboard.SetText Label2.Caption

End Sub

Private Sub Command5_Click()

Label3.Caption = Clipboard.GetText

End Sub

Private Sub Command6_Click()

Picture2.Picture = Clipboard.GetData()

End Sub

Private Sub Command7_Click()

Picture2.Picture = LoadPicture

End Sub

Private Sub Command8_Click()

Text1.Text = Trim(Text1.Text)

If UCase(Right(Text1.Text, 3)) = "BMP" Then

Check1.Value = 1

Else

Check1.Value = 0

End If

If Check1.Value = 1 Then

Clipboard.SetData LoadPicture(Text1.Text), 8

Else

Clipboard.SetData LoadPicture(Text1.Text)

End If

End Sub

Visual Basic - Guía del Estudiante  Cap. 5 Cont.
EL EDITOR DE MENUS

El Menú o Barra de Menú es la segunda línea de cualquier ventana Windows, donde pueden verse distintos nombres. La Barra de Menú es un componente de un Formulario.

La Barra de Menú puede tener tantas palabras como se desee, y sobre cada palabra, puede desplegarse un Menú desplegable con mas palabras. Sobre cada una de estas puede a su vez desplegarse otro conjunto de palabras,  etc. con lo que se puede conseguir una cantidad de palabras tan grande como necesitemos. 

       


   


      


Para cada palabra se produce el evento click. Cada palabra de la Barra de Menú lleva anexo un único procedimiento, que se ejecutará al hacer click sobre la palabra.  Pero en las palabras que sirven para desplegar un menú no tiene sentido que se ponga ningún código en su procedimiento, ya que están ahí justamente para desplegar otras palabras. Será en las palabras finales donde se colocará el código correspondiente.

En las figuras anteriores pueden verse tres formas distintas de ver el mismo Menú. En la primera figura el Menú está sin desplegar.  En la segunda figura puede verse un árbol de Menú largo, desplegado en su totalidad.  En la tercera puede verse el Menú desplegable de la palabra Archivo de la Barra de Menú.  Puede apreciarse en este menú desplegable una línea que separa las palabras Guardar Como  y  SALIR.  Esta línea separadora es muy práctica para separar dos temas distintos dentro de un Menú desplegable. Tan distintos como las operaciones de Abrir y Guardar respecto a la última orden del menú desplegable que es SALIR de la aplicación.

Para crear un menú debe usarse el Editor de Menú, que se encuentra en la Barra de Menú de VB en Herramientas | Editor de menús.  Le aparecerá el siguiente cuadro.  (En principio vacío)






Cada palabra del menú tiene su Caption  (La palabra que aparece en el menú), su Name (Nombre), y puede tener Index (si existen varias palabras con el mismo Name).  Para introducir estas propiedades basta con teclearlas directamente en cada una de las casillas al efecto del cuadro del editor de menús. Para colocar el cursor de escritura en una u otra casilla, puede hacerse bien con el cursor del ratón, bien con el tabulador. Para pasar a la siguiente línea de edición (crear una nueva palabra), basta con pulsar ENTER.

Existen otras propiedades aparte del Caption y Name que se introducen directamente en las casillas al efecto. 

HelpContextID
Igual que esta propiedad en el resto de los controles, permite introducir un número de contexto para la ayuda de Windows.

Enabled. 
Habilitado.  Si se quita la marca de esa casilla, esa palabra aparecerá por defecto deshabilitada. Esta propiedad puede leerse y cambiarse en tiempo de ejecución, por lo que esta propiedad puede usarse para deshabilitar ciertas funciones de la aplicación, dependiendo de la evolución de la propia aplicación.  La presentación en el menú cuando está deshabilitada es con color pálido.

Visible.

Como su nombre indica, que sea visible o no lo sea. Puede cambiarse en tiempo de ejecución.

WindowList
Indica si esa palabra es la que va a mostrar todos los formularios abiertos en una aplicación de documentos múltiples  (MDI).  Al hacer click sobre esta palabra, se desplegará un submenú con todos los Caption de los Formularios hijo abiertos en ese momento. Solamente puede existir una palabra en un menú con esta propiedad activada. Esta propiedad no puede variarse en tiempo de ejecución.

Para cambiar una de estas propiedades en tiempo de ejecución, basta con citar por su nombre a esa palabra (por su Name, no por su Caption), e igualar a True o False su valor :


nmuDocumentos.Visible = False

El nombre usado para nombrar a una de las palabras del menú puede ser cualquiera. Sin embargo es prudente usar un nombre que nos pueda identificar, por una parte, que ese nombre corresponde a una de las palabras del menú, y por otra, a que palabra se refiere.  La solución puede ser cualquiera. El autor de estos apuntes propone poner las letras mnu  (o menu  - recomiendo no usar acentos e los Names de VB), seguido del Caption de esa palabra.

Otras propiedades

Shortcut
Atajo. Esta propiedad admite varias combinaciones de teclas para acceder al procedimiento click de esa palabra sin necesidad de usar el ratón. Las combinaciones de teclas posibles se muestran haciendo click en la flecha hacia abajo que tiene esa casilla. Se desplegará una lista con todas ellas. Para elegir una de estas combinaciones, basta con hacer click sobre la línea que la contiene. Esta combinación de teclas aparecerá en el menú, al lado de la palabra  (Véase figura siguiente,  SALIR    Ctrl + X)

NegotiatePosition.
Establece un valor que determina si los controles Menu de nivel superior se muestran en la barra de menús mientras un objeto vinculado o incrustado de un formulario está activo y mostrando sus menús. No está disponible en tiempo de ejecución. Puede tener los valores 0 (Predeterminado) Ninguno. El menú no se muestra en la barra de menús cuando el objeto está activo.

1  Izquierda. El menú se muestra a la izquierda de la barra de menús cuando el objeto está activo.

2  Medio. El menú se muestra en medio de la barra de menús cuando el objeto está activo.

3  Derecha. El menú se muestra a la derecha de la barra de menús cuando el objeto está activo.

El menú editado con los datos de la figura anterior tendrá esta forma en tiempo de ejecución :





Observe que tiene cuatro palabras en la línea superior, y que “colgando” de la primera aparece un menú desplegable.  Para conseguir un menú desplegable fíjese en la figura del Editor, en la parte de abajo donde están todos los Caption de las palabras del menú. Observará que algunas de ellas  (&Edicion, Documentos, Ayuda  - Hay otra que no se vé, &Archivo) están completamente a la izquierda de la ventana,  y las demás están separadas de la parte izquierda por unos guiones. Las palabras que están completamente a la izquierda son las que figurarán en la barra de menú de forma permanente. Las que están separadas (tabuladas) corresponden a las que aparecerán en los menús desplegables.  Como es lógico, un menú desplegable debe colgar de una palabra de las de la barra de menú. La palabra de la que cuelga el submenú es la palabra sin tabular inmediatamente anterior. 

Para tabular las palabras, hay que hacer click en el botón en forma de flecha a la derecha que está sobre la ventana de edición :





Situandonos sobre una de las líneas ya editadas, y haciendo click sobre el botón con flecha hacia la izquierda, quitamos la tabulación.

Los botones con flecha hacia arriba o abajo nos permiten variar el orden de las líneas ya editadas.

Podemos hacer varios niveles de submenús a base de introducir varias tabulaciones. Al comienzo de este capítulo puede ver un ejemplo de ello.

Quedan un par de cosas. Una de ellas es el carácter & que está delante de los Caption de Archivo y Edición. Este carácter nos permite acceder al procedimiento click de esa palabra pulsando la tecla Alt  +  la tecla correspondiente a la letra que está detrás del carácter &.  Justamente la letra que está subrayada en la palabra del menú.

El carácter & no tiene porqué ir precisamente al comienzo de  la palabra. Puede ir en el medio de la palabra. De esta forma se subrayará una letra intermedia.

Otra cosa. Puede verse en el menú desplegado de la figura anterior una línea entre Guardar Como y SALIR.  Esta línea, que solamente tiene efectos estéticos, se logra introduciendo como Caption un guión  ( - ).  El editor de menús solamente le permite el guión en una palabra de menú desplegable (Tabulada). No se olvide de poner el Name incluso para este guión.

NOTA MUY IMPORTANTE
Siempre se recomienda poner la orden de SALIR en la primera palabra de la barra de Menú. Bien que esa primera palabra sea Salir o que esté en el menú desplegable de la primera palabra de la Barra de Menú, en este último caso, separada mediante una línea separadora. (Vea The Windows Interface Guidelines for Software Design, pág.124) 

POPUP  MENÚS
Un PopUpMenú o Menú Emergente es un menú que despliega en cualquier punto de un Formulario (MDI o normal) con varias opciones. Es normal en los procesadores de texto que se obtenga un menú emergente pulsando el botón derecho del ratón, donde aparecen las tres palabras mágicas de Copiar, Cortar y Pegar.

Para crear un menú emergente o PopUpMenu es necesario tener en el menú de ese formulario una palabra de la que se despliegue un submenú con las mismas palabras que queremos que aparezcan en el PopUpMenu.   

Por ejemplo, en el menú que hemos editado como ejemplo anteriormente, existía una palabra Edición, de la que se desplegaba el Submenú Copiar, Cortar y Pegar.  Si analizamos la edición realizada, las palabras tenían el siguiente Caption y Name :


Caption
Name

Tabulada


Edición
mnuEdicion
NO


Copiar
mnuCopiar
SI


Cortar
mnuCortar
SI


Pegar
mnuPegar
SI

Recuerde que las palabras del menú que estaban tabuladas son las que aparecían en el menú desplegable. Esas mismas serán las que aparezcan en el PopUpMenu.

Para ello utilizaremos el Método PopupMenu. Este método pesenta un menú emergente en un objeto MDIForm o Form en la posición actual del mouse o en las coordenadas especificadas.

La sintaxis de este método es la siguiente :


NombreFormulario.PopupMenu nombremenú, indics, x, y, negrita

Donde NombreFormulario es el nombre del formulario donde presentamos el PopUpMenu. Puede omitirse.

Nombremenú. Este parámetro es requerido. Es el Name de la palabra del menú de donde cuelga el menú desplegable. La palabra del menú a que se refiere Nombremenú debe tener al menos un submenú.

Indics. Parámetro opcional. Un valor o constante que especifica la posición y el comportamiento del menú emergente, como se describe a continuación :


Constante (posición)
Valor
Descripción


vbPopupMenuLeftAlign
0
(Predeterminado) El lado izquierdo del menú se






sitúa en x.


vbPopupMenuCenterAlign
4
El menú emergente se centra en x.


vbPopupMenuRightAlign
8
El lado derecho del menú se sitúa en x.


Constante (comportamiento)
Valor
Descripción


vbPopupMenuLeftButton

0
(Predeterminado) Los elementos del







menú emergente sólo reaccionan a los 







clic del mouse cuando se use el botón 







primario del mouse.


vbPopupMenuRightButton

2
Los elementos del menú emergente







reaccionan a los clic del mouse cuando se 







usen los botones primario o secundario. 

Para combinar la los valores de indics de posición y comportamiento basta con sumar los valores.

Nota   El parámetro indics no tiene efecto en las aplicaciones que se ejecuten bajo Microsoft Windows versión 3.0 o anteriores. 

X, y  Parámetro(s) opcionales que especifican las coordenadas x e y donde va a aparecer el menú emergente. Si no se especifican, aparecerá donde esté el puntero del mouse. Las unidades de medida de las coordenadas x e y se definen mediante la propiedad ScaleMode.

negrita. Parámetro opcional que especifica el nombre de un control menú del menú emergente para presentar su título en negrita. Si se omite, ningún control del menú aparece en negrita. 

Nota.  Este argumento funciona solamente en las aplicaciones que se ejecuten bajo Windows 95. La aplicación ignora este argumento cuando se ejecute bajo versiones de 16 bits de Windows o Windows NT 3.51 y anteriores.

La forma mas habitual de presentar un menú emergente es mediante el botón derecho del ratón. Concretamente en el editor Word que estoy utilizando para escribir estos apuntes ocurre cuando se levanta el botón derecho del ratón. Es decir, en el evento MouseUp, con la condición de que Button=2.  Imaginemos que queremos presentar en un menú emergente las palabras Cortar, Copiar y Pegar del ejemplo anterior, que colgaban en el menú de la palabra Edición. (Vea mas atrás). Queremos también que la palabra Copiar salga resaltada en negrita.  Iríamos al procedimiento MouseUp del Formulario y escribiríamos :

Private Sub Form_MouseUp(Button As Integer, Shift As Integer, X As Single, Y As Single)

If Button = 2 Then

PopupMenu mnuEdicion, , , , mnuCopiar

End If

End Sub

Observe que se ha omitido el nombre del Formulario (Opcional), el parámetro indics, y las coordenadas x e y (el menú emergente aparecerá donde esté situado el cursor del ratón). Observe también que las comas separadoras de los parámetros hay que colocarlas.

Cuando se presenta un menú emergente, el código que sigue a la llamada al método PopupMenu no se ejecuta hasta que el usuario elige un comando del menú (en cuyo caso se ejecuta antes el código del procedimiento de evento Click) o cancela el menú. Además, sólo puede presentarse un menú emergente al mismo tiempo; por lo tanto, las llamadas a este método se ignoran si el menú emergente está presentado actualmente o si un menú desplegable está abierto.

Si queremos presentar un menú emergente es necesario tener ese menú en el menú del formulario. El problema puede ser que no queramos que esté en la barra de menú. No es problema. Ponga la propiedad Visible del elemento del menú desde el que se va a desplegar el submenú a False. No se verá ya en la barra de menú ni esa palabra ni el submenú que de ella se despliega. Eso sí, las palabras del submenú deben tener la propiedad Visible = True. Si les pone esa propiedad a False, no se verán en el menú emergente. Puede aprovechar esta circunstancia para hacer menús emergentes con mas o menos elementos, según las necesidades de su aplicación. También puede poner la propiedad Enabled a False si así lo precisa.  No puede activar la propiedad WindowList en un elemento que forma parte de un menú emergente.

